The Filter Foundry

Revision: 14 September 2021 by Daniel Marschall

This manual is a modified version of the original Filter Factory manual, extended by specific information of Filter Factory as well as by undocumented functions which were present in Filter Factory.

Table of Contents


Introduction	2
Downloading and installing Filter Foundry	3
Windows version 1.7	3
Non-RGB modes Creating custom filters	
To create a custom filter To save a custom filter for additional use Protecting filters To edit a custom filter Expressions for creating filters	
About digital images Components of expressions Number constants Variables Functions Operators Providing user-controlled sliders	6 7 11
Affecting a single channel (filter)	14 15 15
YUV color space variables Polar coordinate system D, dmin, dmax	16 16 17 17
License	10

Introduction

Filter Foundry is a compatible replacement for **Filter Factory** (developed by the Adobe employee Joe Ternasky in 1994). Initially written by Toby Thain (<u>Telegraphics</u>) in 2003 - 2009, the development has been continued by <u>Daniel Marschall</u> (<u>ViaThinkSoft</u>) since 2018. Several advancements and improvements have been made, and a 64-bit Windows version was created.

With Filter Foundry, you can create your own plugins by determining how you want the filter to affect the channels (red, green, blue, and alpha) of each pixel in the image by specifying arithmetic expressions.

The filters you create can also include Settings dialog boxes, which provide up to eight sliders for adjusting the



filter's effect. When you design a filter, you include user-supplied slider information in the expression. You also determine the number of sliders and whether they appear in the Settings dialog box individually or in pairs.

When you create a filter, you can save its expressions in a text file ("AFS file"). Doing so lets you use the Filter Foundry to edit the expressions later. Filter Foundry can also extract the expressions out of standalone plugins created by Filter Factory or Filter Foundry, as long as the plugin is not protected by the creator.

The plugin is free and can be downloaded at:

https://www.viathinksoft.com/projects/filter_foundry


Since the plugin is licensed under the license terms of the GNU General Public License, Version 2, you can download the source code and modify it:

https://github.com/danielmarschall/filter foundry/

More information about the Filter <u>Factory</u> can be obtained from The Filter Factory Programming Guide by Werner D. Streidt and Harald Heim. The guide is for Filter Factory, but most parts also apply to Filter Foundry.

https://thepluginsite.com/knowhow/ffpg/ffpg.htm

The following picture shows an example of a filter created using only a few keystrokes!



Downloading and installing Filter Foundry

Windows version 1.7

Filter Foundry 1.7 comes with a 32-Bit Windows plugin (FilterFoundry.8bf) and a 64-Bit Windows plugin (FilterFoundry64.8bf).





FilterFoundry.8bf

FilterFoundry64.8bf

To install the plugin to Photoshop, simply place the appropriate 8BF file into the **Plug-Ins\Filters** subdirectory of your Adobe Photoshop program files path and restart Photoshop. For other host applications like GIMP, look at the manual on how to install ".8bf" Photoshop filters. If you have information on whether or whether not Filter Foundry works on a non-Photoshop host application, please send us a message!

Macintosh version 1.6

The Macintosh version could not be taken over because Apple removed the "Carbon" API, and the new "Cocoa" API is not compatible with the current code-base. If you would like to help porting Filter Foundry to the latest OS X version, it would be highly appreciated! An old version of Filter Foundry for Mac 68k (requires 68020 or later CPU and FPU) and Mac Classic (PowerPC) can be obtained here: https://www.telegraphics.com.au/sw/

Compatibility

Filter Foundry works with nearly all programs that can handle ".8bf" Photoshop filters.

Following hosts have been tested with Filter Foundry 1.7:

- Photoshop 3.0.5 (x86) through Photoshop CC 2019 (x64)
- IrfanView 4.53 (32/64 bit)
- JASC PSP 9, Corel's Paint Shop Pro XI
- The Gimp 2.2 with PSPI.exe extension to run Photoshop .8bf files
- Serif PhotoPlus 6
- PluginCommander 1.62 (Revision 2)
- Paint.Net using <u>PSFilterPdn</u> to run Photoshop .8bf files

Operating systems:

- Windows NT4, 2000, XP, Vista, 7, 8, 8.1, and 10 are fully compatible (32- and 64-bit)
- Windows 95, 98, Me: Supported, but 64-bit standalone filters cannot be created
- Macintosh currently not supported
- Linux (Wine) not tested

Non-RGB modes

Unlike Filter Factory, Filter Foundry can also work with non-RGB mode images.

For example, if you have a CYMK image, the channels in the Graphical User Interface will have the labels "C", "Y", "M", and "K".

However, the formulas still must contain the variables "r" (for Channel 1), "g" (for Channel 2), "b" (for Channel 3), and "a" (for Channel 4).

So, with CYMK, the expression "r" would refer to the Cyan channel, "b" refers to the "Yellow" channel, "g" refers to the "Magenta" channel and "a" refers to the "Black/Key" channel. The 5th channel (Alpha-channel) cannot be used in this color mode!

Creating custom filters

The procedures in this section explain how to use the Filter Foundry to apply and save filters. For a complete discussion of using arithmetic expressions to achieve an effect, see "Expressions for Creating Filters".


To create a custom filter

- 1. Choose **Telegraphics > Filter Foundry...** from the Filter menu. The Filter Foundry Settings dialog box appears.
- 2. Specify an expression for each channel in the channel fields. Even if you specify the same expression in all three channels ("R", "G", "B"), their evaluations will probably be different. The dialog box will include channel fields for any alpha channels in the image. If you are working in a layer, a channel field ("A") will appear in the dialog box.

For information on how to use expressions to achieve a result, see "Expressions for Creating Filters".

As you type an expression, a small yellow caution sign appears. It will remain visible until you have typed a legal expression. If the caution sign does not disappear, it means that there is an error in the expression. To see which part of the expression is in error, click the caution sign to see the error message and select the incorrect portion.

3. If the expressions include user-supplied slider information, drag the appropriate Map sliders to preview the effects. The Map 0 sliders correspond to sliders 0 and 1; the Map 1 sliders correspond to sliders




- 4. When you have correctly set up the filter, click **Save** to save the expressions in a text file ("AFS parameter file"). Saving the expression allows you to load and edit the filter in the future.
- 5. If you want to use this one instance of the filter only, click OK to apply the filter. If you want to use the filter more than once, see the next procedure, "To save a custom filter for additional use."

To save a custom filter for additional use

- 1. Follow steps 1 through 4 of the previous procedure, "To create a custom filter."
- 2. Click Make. The Custom Filter dialog box appears.
- 3. Name the filter using the **Category** and **Title** fields. The name will appear in the Filter menu under the submenu specified in the Category field.
- 4. Use the **Copyright** and **Author** fields to include credits or copyright information in the filter's **About...** dialog box.
- 5. If the filter's expressions include user-supplied slider information, select the appropriate number of Control or Map options and specify labels for the sliders in the corresponding text boxes. The labels will appear with the sliders in the filter's Settings dialog box.


To display the sliders individually in the Settings dialog box, use the Control options. To display the sliders in pairs, use the Map options. Whether you should use individual or paired sliders depends on the type of filter you are creating and the expressions you have written.

6. Click **OK**. A standard Save dialog box appears. Save the filter somewhere on your disk and/or publish it accordingly.



Filter Foundry will automatically create a 32-bit and a 64-bit filter for you. The 64-bit filter will always have the suffix "64". For example, if you name your filter "Example", then two files will be created: "Example.8bf" (32 bit) and "Example64.8bf" (64-bit).

- 7. Copy the appropriate 8BF file into the Adobe Photoshop Plug-Ins folder.
- 8. To make the filter available to users, restart the Adobe Photoshop program.



Protecting filters

In the step of creating a standalone filter, you can select the option "**Protect and obfuscate**". This will encrypt the filter formulas in the plugin (to avoid that the formulas are read by hex-editors or other tools) and protect it to avoid that it can be opened using Filter Foundry.

Attention: Please make sure that you save your filter in an AFS parameter file since you will NOT be able to recover the formulas from a protected 8BF plugin file! Note that it is technically possible to decrypt a protected filter; however, it is very complex and needs software development skills. The plugin author did his best to make the protect feature as secure as possible and did not create an unprotect/decrypt tool.

To edit a custom filter

- 1. Choose **Telegraphics > Filter Foundry...** from the Filter menu. The Filter Foundry Settings dialog box appears.
- 2. Click Load. Use the Open dialog box to load a file. You can open the following file types:

Extension	Description	
AFS	Parameter file created by Filter Factory or Filter Foundry for Adobe Photoshop	
8BF	Adobe Photoshop plugin	
	It must have been created by Filter Factory or Filter Foundry and must be unprotected.	
PFF	Parameter file created by Filter Factory or Transaction Factory for Adobe Premiere	
PRM	Adobe Premiere plugin	
	It must have been created by Filter Factory or Transaction Factory for Adobe Premiere.	

3. Follow the steps in the previous two procedures to edit and remake the filter.

Expressions for creating filters

This section explains how to set up the arithmetic expressions that describe what a custom filter will do.

About digital images

A digital image is a conglomeration of tiny picture elements, called pixels. Pixels project the color and brightness of the image. Each pixel in an image is uniquely identified by its coordinates. The first coordinate is the horizontal position of the pixel ("x"), and the second coordinate is the vertical position of the pixel ("y"). The horizontal coordinates start counting at the left edge of the image and increase as you move to the right. The vertical coordinates start counting at the top of the image and increase as you move down. Therefore, the top left corner of the image has the coordinates (0,0). The range of coordinates for an image depends on its resolution.

In **RGB color mode**, the color of a pixel is stored as three numbers: the amount of red, the amount of green, and the amount blue. The three-color values are called *channels*. Channel values can range from 0 to 255. Pixels can have a fourth channel, an alpha channel, which controls the transparency of the image.

- sivel
- 1. If a channel value is set to 0, none of its colors is present in the pixel.
- 2. If a channel value is set to 255, the maximum amount of that color is present in the pixel. For example, if a pixel has the channel values (255,0,0), the pixel is entirely red: 255 red, 0 green, 0 blue.
- 3. If all three channels have the same value, the pixel is a shade of gray. For example, (80,80,80) is a dark gray, (128,128,128) is a medium gray, and (200,200,200) is a light gray.
- 4. If all three channels are 0, the pixel is black. If all three channels are 255, the pixel is white.

The filters you create affect the channel values of the pixels in an image. You specify an expression for each channel, and each operation is performed on the appropriate channel for every pixel in the image. Expressions can include specific pixel coordinates whose channel values are evaluated and used in the calculation.

Note: If an expression evaluates to a number greater than 255, the channel is set to 255. Likewise, if an expression evaluates to a number less than 0, the channel is set to 0.

Components of expressions

A filter performs an operation on the channels of each pixel in an image. These channel operations are described by arithmetic expressions. Expressions are made up of combinations of four types of components: number constants, variables, functions, and operators. The following sections describe these four components.

The Filter Foundry allows only integer numbers in expressions – no fractions or decimal numbers are allowed. Variables and functions will always evaluate to integers.

Number constants

A number constant is a number that is supplied directly in the expression. Constants can be used to construct simple expressions such as 10+5, and these expressions can always be replaced by another constant; in this case, 15.

Constants can also be written in hexadecimal form. To use hexadecimal values, prefix the number by 0x, as in 0xff which is the same as 255 in decimal notation.

Variables

A variable is a short name, such as x, that can be evaluated. The value of a variable can depend on the current image, the current pixel, or the current channel for which an expression is being evaluated. For example, the variable x always evaluates to the horizontal coordinate of the current pixel, and the variable y always evaluates to the vertical coordinate. The variables for the current pixel's channel values are r (red), g (green), b (blue), and optionally a (alpha).

You can combine variables and constants to form expressions. For example, the expression r+g retrieves the red and green channel values for the current pixel and adds them together.

You can use the following variables in your expressions:

Variables	Definitions
r, g, b	Red, green, and blue channel values for the current pixel
a	Alpha channel (transparency) value for the current pixel. 0 (amin) is fully transparent, 255 (A, or amax) is fully opaque. Does not work on the background layer or layers where transparency is locked.
С	Value of the current channel, whichever channel the expression is defining
R, G, B, A, C (rmax,)	Maximal channel values. Are always 255.
rmin, gmin, bmin, amin, cmin	Minimum channel values. Are always 0.
i, u, v	Calculated channel values for the current pixel in YUV color space
imin	Smallest possible value of <i>i</i> . It is always 0.
umin	Smallest possible value of <i>u</i> . It is always -55. In Filter Factory, this variable had the value 0 (bug?).
vmin	Smallest possible value of v. It is always -78. In Filter Factory, this variable had the value 0 (bug?).
imax	Largest possible value of <i>i</i> . It is always 255.
umax	Largest possible value of <i>u</i> . It is always 55. In Filter Factory, this variable had the value 255 (bug?).
vmax	Largest possible value of v. It is always 78. In Filter Factory, this variable had the value 255 (bug?).
I	Defined as imax-imin=255
U	Defined as umax-umin=110
V	Defined as vmax-vmin=156
d	Direction (angle) of the current pixel from the center of the image, where d is an integer between -512 and 512, inclusive
dmax	Max value of d. It is always 512 (In Filter Factory, it was erroneously 1024)

dmin	Min value of d. It is always -512 (In Filter Factory it was erroneously 0)
D	Returns the total amount of angles within the image. Is always 1024 (dmax-dmin)
m	Distance (magnitude) from the center of the image to the current pixel
M (mmax)	Range of magnitudes with the image. It is always one half the diagonal size of the image
mmin	Is always 0
х	X-Coordinate (horizontal) of the current pixel
X (xmax)	Range of horizontal coordinates over the width of the image
xmin	Is always 0
у	Y-Coordinate (vertical) of the current pixel
Y (ymax)	Range of vertical coordinates over the height of the image
ymin	Is always 0
z	Channel index for the current expression
Z (zmax)	Range of channel indexes within one pixel
zmin	Is always 0

Note: Some of these variables were present in Filter Factory, but they were undocumented! The variables in brackets are synonyms.

5. The i, u, and v variables do not exist in an RGB image, so they are calculated from the RGB channels. Because this calculation takes some time, using these variables is slower than using the r, g, and b variables. The following formulas are used to convert from RGB to YUV:

```
i = (299*r + 587*g + 114*b) / 1000

u = (-147407*r - 289391*g + 436798*b) / 2000000

v = (614777*r - 514799*g - 99978*b) / 2000000
```

According to these formulas, the output range of i, u, and v is:

 $0 \le i \le 255$

 $-55 \le u \le 55$

 $-78 \le v \le 78$

6. The X variable represents the largest possible value for the x variable. The Y Variable represents the largest possible value for the y variable. The Z variable represents the largest possible value for the z variable. These ranges are closed on the minimum and open on the maximum: $0 \le x < X$, $0 \le y < Y$, and $0 \le z < Z$.

Functions

Functions are short names that can be evaluated, and they require one or more arguments.

For example, the rnd function requires two arguments. It evaluates to a number that is greater than or equal to the first argument and less than or equal to the second argument. The expression rnd(1,10) evaluates to a number between 1 and 10, inclusive. (The rnd function is called a random number generator, and it is useful for adding noise or texture to an image.)

Arguments are written within parenthesis and are separated by commas. The arguments can be expressions. For example, the expression rnd(r-10,r+10) evaluates to the red channel of the current pixel, plus or minus 10.

Another function, src (source), retrieves channel values for a particular pixel. It requires three arguments: the horizontal coordinate of the pixel, the vertical coordinate, and the channel index. (The index for the red channel is 0; the green channel is 1; the blue channel is 2; the alpha channel is 3.) For example, the expression src(10,20,0) retrieves the red channel value for the pixel at coordinates (10,20). The expression src(x,y,0) retrieves the red channel value for the current pixel. The expression src(x+1,y,0) retrieves the red channel for the pixel to the right of the current pixel.


You can use the following functions in your expressions. Many functions place restrictions on the possible values of their arguments. If an argument is out of range, the expression will return a 0. For example, the expression *ctl*(8) evaluates to 0 because the *ctl* function requires an argument between 0 and 7.

Functions	Definitions	
src(x,y,z)	Channel z (0 red, 1 green, 2 blue, 3 alpha) for the pixel at coordinates x, y	
rad(d,m,z)	Channel z in the source image, which is m units away, at an angle of d , from the center of the image	
ctl(i)	Value of slider <i>i</i> , where <i>i</i> is an integer between 0 and 7, inclusive	
val(i,a,b)	Value of slider i , mapped onto the range a to b	
map(i,n)	Item n from mapping table i , where i is an integer between 0 and 3, inclusive, and n is an integer between 0 and 255, inclusive	
min(a,b)	Lesser of a and b	
max(a,b)	Greater of a and b	
abs(a)	Absolute value of <i>a</i>	
add(a,b,c)	Sum of a and b , or c , whichever is lesser	
sub(a,b,c)	Difference of a and b , or c , whichever is greater	
dif(a,b)	Absolute value of the difference of a and b	
rnd(a,b)	Random number between a and b, inclusive	
rst(i)	Resets the random-number-generator using seed i (032767). The RNG is also reset every time the filter is invoked. The initial seed is 0.	
mix(a,b,n,d)	Mixture of a and b by fraction n/d , $a*n/d+b*(d-n)/d$	
scl(a,il,ih,ol,oh)	Scale a from input range (il to ih) to output range (ol to oh)	
sqr(x) or sqrt(x)	Square root of x	
	Filter Factory did calculate $sqr(x)=x$ for the illegal input value $x<0$. Filter Foundry took over this behavior to keep compatibility.	
$\sin(x)$	Sine function of x, where x is an integer between 0 and 1024, inclusive, and the value returned is an integer between -512 and 512, inclusive (Windows) or -1024 and 1024, inclusive (Mac OS)	
cos(x)	Cosine function of x, where x is an integer between 0 and 1024, inclusive, and the value returned is an integer between -512 and 512, inclusive (Windows) or -1024 and 1024, inclusive (Mac OS)	
tan(x)	Tangent function of x , where x is an integer between -256 and 256. The value returned can go down to $-\infty$ or up to ∞ . (The Filter Factory documentation described this function to be bounded, but this is not true).	
r2x(d,m)	x displacement of the pixel m units away, at an angle of d , from an arbitrary center	
r2y(d,m)	y displacement of the pixel m units away, at an angle of d , from an arbitrary center	
c2d(x,y)	Angle displacement of the pixel at coordinates x,y	
c2m(x,y)	Magnitude displacement of the pixel at coordinates <i>x</i> , <i>y</i>	
get(i)	Returns the current cell value at i	
put(v,i)	Puts the new value v into cell i	
cnv(m11,m12,m13, m21, m22,m23, m31, m32, m33, d)	Convolution kernel where $m11-m33$ define the weights of pixels surrounding the current pixel $(m22)$ and d scales the resulting value	
pow(b,e)	Calculates the base to the exponent power, that is, b^e . (This function was added in the <u>inofficial Filter Factory patch 3.1.0</u> by Daniel Marschall.)	

- 7. The *src* (source) function is slow compared to the other operators and functions. Several evaluations of the *src* function in one expression can noticeably slow down the processing of an image. The coordinates passed to the *src* function should be within the ranges specified by the *X*, *Y*, and *Z* variables; otherwise, the coordinates will be pinned.
- 8. The *val* (value) function converts the requested slider setting into a value in the requested range. The slider setting is multiplied by the size of the range (b-a) and offset by the start of the range (a). This function is useful when the range returned by the sliders (always 0 to 255, inclusive) does not match the range of values you want to use. For example, if the requested range is 1 to 10, a slider setting of 0 returns a value of 1, a setting of 255 returns a value of 10, and a setting of 127 returns a value of 5. The start of the requested range does not have to be less than the end of the range. For example, the expression val(0,10,-10) returns values between 10 and -10.
- 9. The map (mapping) function uses tables that are constructed according to the slider settings. Each table uses a pair of sliders: table n uses sliders 2n and 2n+1 for the high and low values, respectively. The table is constructed as follows, where L is the value of the low slider, H is the value of the high slider, and I is an entry:

```
if I \le L, use 0;
if I \ge H, use 255;
if L \le I \le H, use (I - L) *255/(H - L).
```

- 10. The *rnd* (random) function returns a different random number each time it is called, but the entire function resets each time an image is processed. As a result, a filter that uses the *rnd* function will have the same effect each time it is used on the same image.
- 11. The *rst* (random state) function can be used to change the initial seed of the random-number-generator (*rnd* function). More information on how *rst* works in Filter Factory and Filter Foundry can be found in the chapter "Implementation detail differences".
- 12. The *mix* (mixture) function combines the two input values using the specified fraction. A fraction of 1/2 returns the average of the two input values. A fraction close to 1 returns the first input value, and a fraction close to 0 returns the second input value. The mix of (a,b,n,d) is defined as a*n/d+b*(d-n)/d.
- 13. The scl (scale) function maps a value from an input range onto an output range. For example, an input range of 0 to 255 could be mapped onto an output range of -100 to 100 by the expression scl(c,0,255,-100,100). In this example, channel values close to 0 are mapped starting at -100, and channel values close to 255 are mapped up to 100. Note that val(0,a,b) is equivalent to scl(ctl(0),0,255,a,b).
- 14. *sin* and *cos* are bounded while *tan* will go to +/- infinity (actually, the range is -167772 to 167772). The following diagram shows a diagram of the three trigonometric functions:



15. The *r2x* and *r2y* functions convert radial coordinates to cartesian coordinates. The *c2d* and *c2m* functions convert Cartesian coordinates to radial coordinates.

16. The get and put functions can be used to store intermediate values into one of the cells (indexed from 0 to 255). Because the channel expressions are evaluated in order (R,G,B,A), the cells can be filled with values in an expression, and the values are available for the following functions. For example, the following expressions:

R: put(i, 0), get(0)

G: get(0)B: get(0)

A: a

are an efficient way to convert a color image into a monochrome image. Using get and put in this case speeds things up because the expression i is only evaluated once per pixel, rather than for every channel of every pixel.

17. The cnv (convolve) function is like Photoshop's Custom filter and modifies the color of a pixel by assigning weights to the pixel and its eight surrounding pixels. The value of each pixel is multiplied by a weight (m11 though m33). The resulting values are then added and divided by d (typically the sum of the weights) to produce the new pixel value.

The pixels are arranged as follows, where m22 is the weight assigned to the target pixel:

m11	m12	m13
m21	m22	m23
m31	m32	m33

For example, the following expressions blur and sharpen a pixel, respectively:

cnv(0, 1, 0, 4, 1, 0, 1, 0, 8)

			_
0	1	0	
4	1	0	
1	0	8	d=

-8


cnv(0, -1, 0, -1, 5, -1, 0, -1, 0, 1)


0	-1	0
-1	5	-1
0	-1	0

d=1

Polar coordinate functions

Functions in Filter Foundry may use one of two distinct coordinate models to interpret image data. One of these is the Cartesian coordinate model described in "About Digital Images". The other is the polar coordinate model, in which the origin is the center of the image, and coordinates are defined by an angle d from the horizontal axis and a distance, or magnitude, m from the origin. In this model, -512 degrees corresponds to the 9 o'clock position, -256 to the 12 o'clock position, 0 to the 3 o'clock position, 256 to the 6 o'clock position, and 512 to the full rotation back to the 9 o'clock position. Negative values correspond to the appropriate counter-clockwise position. The functions rad, sin, cos, tan, r2x, r2y, c2d, and c2m use polar coordinates.





Note: Filter Factory defines c2d(x,y) := arctan2(y,x), but d := arctan2(-y,-x)

Operators

The operators include all of the arithmetic that can be used in an expression. There are five types of operators:

- 1. Arithmetic
- 2. Relational
- 3. Logical
- 4. Conditional
- 5. Bitwise

Precedence

You can use the following operators in your expressions. The operators are presented in their order of precedence. Precedence determines which operators are evaluated first within an expression when the order of evaluation is ambiguous. For example, in the expression 2+3*4, the * operator is evaluated first because it has higher precedence than the + operator.

You can use brackets to change the precedence of the operators.

Operators	Definitions
,	Sequence
?:	Conditional.
&&,	Logical and, logical or
&, ^,	Bitwise and, bitwise exclusive or, bitwise or
==, !=	Equal to, not equal to
<, <=, >, >=	Less than, less than or equal to, greater than, greater than or equal to
<<,>>>	Shift left, shift right
*, /, %	Multiply, divide, modulo
+, -	Add, subtract
!, ~	Logical not, bitwise not

Arithmetic operators

Operators	Definitions	
a+b	Add	
a–b	Subtract	
a*b	Multiply	
a/b	Divide	
a%b	Modulo	
-a	Negate	

The **arithmetic operators** are +, -, *, /, and %. The % (modulo) operator calculates the remainder of a division. For example, the expression 11%3 evaluates to 2.

The arithmetic operators +, -, *, /, and % operate on signed, 32-bit integers.

Relational operators

Operators	Definitions
a==b	Equal to
a!=b	not equal to
a <b< td=""><td>Less than</td></b<>	Less than
a<=b	less than or equal to
a>b	greater than
a>=b	greater than or equal to

Relational operators compare two expressions and evaluate to 0 (false) or 1 (true). For example, the < operator evaluates to 1 if the expression on the left is less than the expression on the right. The expression r < g evaluates to 1 when the red channel of the current pixel has a lower value than the green channel. Otherwise, it evaluates to 0. The set of relational operators includes <, <=, >, >=, ==, and !=.

The == ("equal-to") operator evaluates to 1 when the two expressions surrounding it evaluate to the same thing. The != ("not-equal-to") operator evaluates to 1 when the two expressions surrounding it evaluate to something different. For example, the expression I == I evaluates to 1, and the expression I == I evaluates to 1, and the expression I == I evaluates to 0.

Logical operators

Operators	Definitions		
a&&b	Logical "and" operation which has the following truth table:		
	a b a&&b		
	0 0 0		
	0 1 1		
	1 0 1		
	1 1 1		
a b	Logical "or" operation which has the following truth table:		
	a b a b		
	0 0 0		
	0 1 0		
	1 0 0		
	1 1 1		
!a	Logical "not" operation which has the following truth table:		
	a !a		
	0 1		
	1 0		

Logical operators let you combine several relational expressions. For example, you could evaluate whether the horizontal coordinate of a pixel is between 10 and 30, inclusive. The appropriate relational expressions are x > = 10 and x < = 30. You can use the logical && operator to combine them into the single expression (x > = 10) && (x < = 30). The &&

operator evaluates the expressions on both sides. If neither expression evaluates to 0, the && operator evaluates to 1. If either expression evaluates to 0, the && operator evaluates to 0.

The logical operators &&, \parallel , and ! treat all expressions as either true or false. Any value other than 0 is considered true, and only a 0 value is considered false.

The logical operator \parallel is similar to the operator &&, but it performs a slightly different logical operation. The \parallel operator is also placed between two relational expressions. If either of the expressions evaluate to anything but 0, the \parallel operator evaluates to 1. If both expressions evaluate to 0, the \parallel operator evaluates to 0. For example, the expression (x>10)||(y>10)| evaluates to 1 when the horizontal coordinate is greater than 10. The only time this expression evaluates to 0 is when the horizontal coordinate is \leq 10 and the vertical coordinate is \leq 10. The following truth table shows the difference between the && and \parallel operators:

Left expression	Right expression	Left && Right	Left Right
0	0	0	0
0	1	0	1
1	0	0	1
1	1	1	1

Finally, you place the ! operator before an expression to invert the expression's evaluation. If the expression evaluates to 0, the ! operator evaluates to 1. If the expression evaluates to anything but 0, the ! operator evaluates to 0.

Conditional operators

Operators	Definitions
a?b:c	Conditional.
	Evaluate "b" if "a" is non-zero, otherwise, evaluate "c".
a,b	Sequence
a,b,c	The comma operator evaluates both the left and right expressions and returns the value of the right expression.

The single **conditional operator**? lets you make a choice between two alternatives. A conditional expression includes a selection expression and two alternative expressions. The conditional operator evaluates the selection expression and uses the result to decide which of the two alternatives it should evaluate. If the selection expression evaluates to anything but 0, the first alternative is evaluated. If the selection expression evaluates to 0, the second alternative is evaluated.

For example, in the expression (x%2)?r:g, the ? conditional operator separates the selection expression (x%2) from the two alternative expressions r and g. The alternative expressions are separated by a colon (:). The selection expression divides the horizontal coordinate of the current pixel by 2 and returns the remainder of the division. If the horizontal coordinate is an odd number, the result is something other than 0, and if the horizontal coordinate is an even number, the result is 0. Therefore, if the pixel has an odd horizontal coordinate, the conditional operator returns the value of the red channel. If the current pixel has an even horizontal coordinate, the conditional operator returns the value of the green channel.

Bitwise operators

Operators	Definitions
a&b	Bitwise "and"
a^b	Bitwise "exclusive or"
a b	bitwise "or"
a< b	Shift left
a>>b	Shift right
~a	Bitwise "not"

The << and >> expressions are placed between two expressions. Both operators shift the bits in the left expression's evaluation by some number, which is specified by the right expression's evaluation. The << operator shifts bits to the left. The >> operator shifts bits to the right.

The shifting operators (<< and >>) perform logical, not arithmetic, shifts so the sign of the shifted operand is not preserved.

Providing user-controlled sliders

When you create a filter, you can provide up to eight slider controls for the user to adjust when applying the effect. Slider values can range from 0 to 255. You set up the effect's sliders by using the *ctl* (control), *val* (value), and *map* (mapping) functions in your expressions.

If you use these functions to retrieve slider information, you should set up the Control or Map options in the Filter Foundry's Build Custom dialog box. For information on using the Build Custom Filter dialog box, see "Creating Custom Filters".

- 1. The *ctl* function retrieves the specified slider's current value. This function requires one argument: the slider index, which is a number between 0 and 7. For example, the expression *ctl(0)* evaluates the current value of the first slider.
- 2. The *val* function converts the range of possible slider values (always 0 to 255) into a range that you specify. For example, to get a value between 1 and 100 from a slider, you would use the expression *val*(0,1,100). If slider 0 is set to 0, the *val* function evaluates to 0. If slider 0 is set to 255, the *val* function evaluates to 100. Slider values between 0 and 255 are converted into values between 1 and 100.
- 3. The map function groups the sliders into pairs. Each even/odd slider pair sets the values in a table, which is accessed by the map function. There are four mapping tables one for each slider pair. Sliders 0 and 1 set the values for mapping table 0; sliders 2 and 3 set the values for mapping table 1, and so on. Each table contains 256 entries, which are calculated each time the slider values change.

The map function takes two arguments: the table index and the item index. For example, the expression map(1,20) returns item 20 from table 1. The table index must be between 0 and 3. The item index must be between 0 and 255, inclusive.

Examples

This section provides several examples of using expressions to achieve a result. The examples are presented in the order of their complexity. The Adobe Photoshop program also provides some sample filter expressions. These samples have been saved as text files and are located in a folder in your Adobe Photoshop folder. You can use the Filter Foundry to load a sample file and observe its effect.

Affecting a single channel (filter)

To make an image redder, you could use the following expressions:

R: r+100 G: g

B: b

1.0

The first expression evaluates the red channel of each pixel and adds 100 to each one. The next two expressions evaluate the other two channels and leave them unchanged.

Affecting channels using sliders (filter)

To add a user-controlled slider value to the current channel values, you could use the following expressions:

R: r+ctl(0)

G: g+ctl(1)

B: b+ctl(2)

A: 0

The first expression evaluates the red channel of each pixel and adds the value of slider 0 to each one. The next two expressions do the same thing to the green and blue channels, using the values of slider 1 and slider 2, respectively.

Adding noise to channels using sliders and random values (filter)

To use slider values to determine the range of random numbers, you could use the following expressions:

```
R: r+rnd(-ctl(0),ctl(0))
G: g+rnd(-ctl(1),ctl(1))
B: b+rnd(-ctl(2),ctl(2))
A: 0
```

The filter defined by these expressions adds noise to all three channels. The amount of noise in each channel is determined by the slider controls. The first expression evaluates the slider setting from slider 0. This value is used as the argument for the rnd function. If the slider setting is 0, the rnd function evaluates to 0. If the slider setting is 100, the rnd function can return any number between -100 and 100, inclusive. The result of the rnd function is then added to the current value of the red channel. As the slider setting is raised from 0 to 255, the random numbers are selected from a wider and wider range, resulting in more and more noise being added to the red channel.

The next two expressions perform the same operation on the green and blue channels, using sliders 1 and 2 for input, respectively.

Amplifying or toning down channels (filter)

To amplify or tone down a channel based on the values of a different channel, you could use the following expressions:

```
R: (b>100) ? r+50 : r-50
G: g
B: b
A: 0
```

The first expression evaluates the blue channel to determine if it is greater than 100. If it is greater than 100, the entire expression evaluates to the red channel value plus 50. If it is not greater than 100, the expression evaluates to the red channel value minus 50. The other two expressions do nothing. Therefore, this filter amplifies the red channel if there is bluer than you want, or it tones down the red channel if there is less blue than you want.

You could also use slider values in a similar type of filter, as follows:

```
R: (b>ctl(0)) ? r+ctl(1) : r-ctl(1)
G: g
B: (b>ctl(0) ? b-ctl(1) : b+ctl(1)
A: 0
```

The first expression uses the setting from slider 0 as a "cutoff" value. If the blue channel is greater than the cutoff value, the red channel is amplified by the value of slider 1. If the blue channel is less than the cutoff value, the red channel is toned down by the value of slider 1. The third expression is the opposite of the first one, but it works on the blue channel instead of the red channel. The effect is that anything that is added to the red channel is subtracted from the blue channel and vice versa.

Averaging the channel values of neighboring pixels (filter)

You can use the *src* (source) function to retrieve the channel values from neighboring pixels and average them together, as follows:

```
R: (src(x-1,y,0)+src(x,y,0)+src(x+1,y,0))/3
G: (src(x-1,y,1)+src(x,y,1)+src(x+1,y,1))/3
B: (src(x-1,y,2)+src(x,y,2)+src(x+1,y,2))/3
A: 0
```

The first expression uses the *src* function to retrieve the red channel value for three different pixels: the pixel to the left of the current pixel, the current pixel, and the pixel to the right of the current pixel. These three values are added together and then divided by 3. The next two expressions do the same thing using the green and blue channels, respectively.

Implementation detail differences

Filter Foundry tries to be as compatible with Filter Factory as possible.

However, there are some differences that are explained in this documentation.

YUV color space variables

Filter Factory:

```
i = ((76*r)+(150*g)+(29*b))/256 Output range is 0 \le i \le 254 u = ((-19*r)+(-37*g)+(56*b))/256 Output range is -55 \le u \le 55 v = ((78*r)+(-65*g)+(-13*b))/256 Output range is -77 \le v \le 77 imin = 0 imax = 255 umin = 0 umax = 255 vmin = 0 vmax = 255 U = 255 U = 255 U = 255
```

Filter Foundry 1.7 uses more accurate formulas which have higher accuracy, as well as correct min/max values which represent the actual range of the *i*, *u*, *v* variables:

```
i = (299*r+587*g+114*b)/1000 Output range is 0 \le i \le 255 u = (-147407*r-289391*g+436798*b)/2000000 Output range is -55 \le u \le 55 Output range is -55 \le u \le 55 Output range is -78 \le v \le 78 imin = 0 imax = 255 umax = 55 vmin = -78 vmax = 78 I := imax-imin = 255 U := umax-umin = 110* V := vmax-vmin = 156*
```

Polar coordinate system D, dmin, dmax

The Filter Factory manual writes:

```
d=0 corresponds to the 3 o'clock position
d=256 to the 6 o'clock position
d=512 to the 9 o'clock position
d=768 to the 12 o'clock position
d=1024 to the full rotation back to the 3 o'clock position
```

But this does not match the Windows implementation of Filter Factory (maybe it is true to the Mac implementation?)

In the original Windows-implementation we can observe:

```
d=-512 is at 9 o'clock position
d=-256 is at 12 o'clock position
d=0 is at 3 o'clock position
d=256 is at 6 o'clock position
d=512 is the full rotation back to 3 o'clock position
```

Therefore, dmin has been changed from 0 to -512, and dmax have been changed from 1024 to 512. It is questionable if D was meant to be a synonym of dmax, or if D was meant to be D := dmax - dmin. We have chosen the latter, so it stayed 1024.

get(i) function

```
Filter Factory: get(x)=x \text{ if } x>255 \text{ or } x<0
Filter Foundry: get(x)=0 \text{ if } x>255 \text{ or } x<0
```

Note: The result "x" was most likely not intended but a result of an undefined behavior

^{*}It is questionable if I was meant to be a synonym of imax, or if I was meant to be I := imax - imin. We have chosen the latter in Filter Foundry 1.7.0.9. Same thing with U and V.

r, g, b at an empty canvas

In Filter Factory, an empty (transparent) canvas of a new file is initialized as 'r=g=b=0'

Filter Foundry initializes it as `r=g=b=255`

rst(i) function

Filter Factory contains an undocumented function that sets the seed for the random number generator. Filter Factory and Filter Foundry beginning with 1.7.0.8 accept a seed between 0 and 32767, inclusively. If the argument is not within this range, the operation lowest 15 bits are taken.

There are many differences in the implementation between Filter Factory and Filter Foundry in regards rst(i), which is documented below.

Filter Factory:

If rst(i) is called in Filter Factory, an internal Seed-Variable is set. It does NOT influence any calls of rnd(a,b), because a lookup-table needs to be built first. The building of the lookup-table is probably done before the processing of the first pixel (x,y,z=0). It is suspected that the call of rst(i) will take effect on the next calculation. Due to a bug (or feature?), the random state is not reset to its initial state (0) before the filter is applied. The preview image processing will modify the random state, and once the filter is actually applied (pressing "**OK**"), the random state that was set in the preview picture, will be used. This could be considered as a bug, but it is probably required, otherwise, the call of rst(i) (inside the preview calculation) won't affect the rnd(a,b) in the real run. However, in a standalone filter without dialog/preview, there is no preview that could set the internal seed, so the rnd(a,b) functions will always work using the default seed 0, and only the subsequent calls will use the rst(i) of the previous call.

Filter Foundry:

In Filter Foundry, the function rnd(a,b) retrieves a random number in "realtime"; therefore, if the seed is changed via rst(i), there is an immediate effect on the next call of the rnd(a,b) function.

For example, the following filter would generate a one-colored picture without any randomness:

```
R: rst(123), rnd(0,255)
G: rnd(0,255)
B: rnd(0,255)
```

If you want to generate a random pixel image with a non-default seed, you need to make sure that rst(i) is called only once at the beginning (channel 0, coordinate 0,0):

```
R: (x== 0 && y ==0) ? rst(123) : 0, rnd(0,255)
G: rnd(0,255)
B: rnd(0,255)
```

In Filter Foundry, rst(i) can be called by branches, and variables/sliders can be used as arguments of rst(i).

Evaluation of conditional branches

Filter Foundry:

Only the branches which will be chosen due to the conditional expression will be evaluated.

This means that following filter would generate a black canvas:

```
R: 1==0 ? put(255,0) : 0
G: get(0)
B: 0
```

In Boolean expressions, the evaluation will be aborted if the result is already determined.

So, this will also generate a black canvas:

```
R: 1==0 && put(255,0) ? 0: 0
G: get(0)
B: 0
```

This will also generate a black canvas:

```
R: 1==1 || put(255,0) ? 0 : 0
G: get(0)
B: 0
```

Filter Factory:

Each branch inside an if-then-else expression will be evaluated. This means that following filter would generate a green canvas:

```
R: 1==0 ? put(255,0) : 0
G: get(0)
B: 0
```

Also, all arguments of a Boolean expression will be fully evaluated.

So, this will also generate a green canvas:

```
R: 1==0 && put(255,0) ? 0: 0
G: get(0)
B: 0
```

This will also generate a green canvas:

```
R: 1 == 1 \parallel put(255,0) ? 0 : 0
G: get(0)
B: 0
```

License

This program is free software; you can redistribute it and/or modify it under the terms of the GNU General Public License as published by the Free Software Foundation; either version 2 of the License, or (at your option) any later version:

https://www.gnu.org/licenses/old-licenses/gpl-2.0.html https://www.gnu.org/licenses/gpl-3.0

This program is distributed in the hope that it will be useful, but WITHOUT ANY WARRANTY; without even the implied warranty of MERCHANTABILITY or FITNESS FOR A PARTICULAR PURPOSE. See the GNU General Public License for more details.